
Introduction: 
R5s140509 raised the issue of support of 5 MHz bandwidth in MAC TBS [7.1.7.x] test cases.  TTCN currently supports bandwidths of 10/15/25 MHz in MAX TBS test cases. Band 31 is a unique case of being only 5 MHz wide and supports 5 MHz as highest bandwidth.  Support is needed in TTCN and MAC TBS sheets for 5 MHz.
Summary of changes:

· 5 MHz column added in 5 TBS sheets

	1
	DCI-1-RA0-ExplicitConfig
	DL Resource scheduling for DCI format 1, Resource allocation 0 and PDCCH is scrambled by C-RNTI

	2
	DCI-1-RA1-ExplicitConfig
	DL Resource scheduling for DCI format 1, Resource allocation 1 and PDCCH is scrambled by C-RNTI

	3
	DCI1A-ExplicitConfig
	DL Resource scheduling for DCI format 1A, Resource allocation 2(localised & distributed) and PDCCH is scrambled by C-RNTI

	4
	DCI-2A-RA0-ExplicitConfig
	DL Resource scheduling for DCI format 2A, Resource allocation 0 and PDCCH is scrambled by C-RNTI

	5
	DCI-2A-RA1-ExplicitConfig
	DL Resource scheduling for DCI format 2A, Resource allocation 1 and PDCCH is scrambled by C-RNTI


· TTCN changes to functions f_EUTRA_7_1_7_1_x and f_EUTRA_7_1_7_1_x_MIMO as highlighted below.
	function f_EUTRA_7_1_7_1_x (PdcchDciFormat_Type p_Format,

                              PdcchResourceAllocation_Type p_ResourceAllocType) runs on EUTRA_PTC

  { /* L2/MAC/7.1

     * 7.1.7.x : DL-SCH Transport Block Size selection

     */

    var integer v_UE_Category ;

    var integer v_TBSize:=0;    // transport block size in bits

    var integer v_I_MCS:=0;     // Modulation and Coding Scehme Index

    var integer v_I_TBS:=0;     // Transport Block Size Index

    var integer v_N_PRB:=1;     // Number of physical resource Blocks

    var integer v_N_RBDL := 100; // Specifies the DL Bandwidth.

    var integer v_Max_nPRB;     // to store the max allowed PRB resources that can be allocated

    var integer v_PDCPsize:=0;  // PDCP SDU Size

    var integer v_N:=1;         // Number of PDCP SDU

    var EUTRA_CellInfo_Type v_CellInfo;

    var PDCP_SDUList_Type v_PDCP_SDUList := {''O};

    ///////////////////////////////////////////////////////////////////////////////////

    //      PREAMBLE

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_Init (c1);

    v_CellInfo:= f_EUTRA_CellInfo_Get(eutra_Cell1);

    f_EUTRA_CellInfo_SetHighestBandwidth(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    v_UE_Category := f_EUTRA_MobileInfo_GetUECapability_UECategoryExtn(); //@sic R5s120641 R5-126021 sic@

    f_CheckUECategory(v_UE_Category);

    // Bring UE to state Loopback Activated (state 4)

    f_EUTRA_LoopBackActivation_State4_717(); //@sic R5-100263 R5s100155 R5s110697 sic@

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1,

                                cas_CellModify_REQ_PDSCH_Pwr_Ratio_TBS_Test(eutra_Cell1,

                                                                            cs_TimingInfo_Now));

    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_TestBody_Set(true);

    //Steps 1 to 4 are repeated for values of nPRB  from 1 to v_Max_nPRB and iMCS from 0 to 28

    v_N_RBDL := f_ConvertDL_BandwidthToInteger(f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1));

    if ((p_Format == dci_1) and (p_ResourceAllocType == ra_0))

      {

        // test case 7.1.7.1.1

        v_Max_nPRB := v_N_RBDL;

        // MAX index can be NRBDL but not all nPRB's allowed

      }

    else if ((p_Format == dci_1) and (p_ResourceAllocType == ra_1))

      {

        // test case 7.1.7.1.2



//5MhzSupport

        if (v_N_RBDL == 25)

          {

            v_Max_nPRB := 11;

          }else if (v_N_RBDL == 50)

          {

            v_Max_nPRB := 14;

          }

        else if (v_N_RBDL == 75)

          {

            v_Max_nPRB := 16;

          }

        else if (v_N_RBDL == 100)

          {

            v_Max_nPRB := 22;

          }

        else

          {

            FatalError(__FILE__, __LINE__, "Step 1: invalid MAX Bandwidth set");

          }

      }

    else if ((p_Format == dci_1A) and (p_ResourceAllocType == ra_2_Localised))

      {

        // test case 7.1.7.1.3

        v_Max_nPRB := v_N_RBDL;

      }

    else if ((p_Format == dci_1A) and (p_ResourceAllocType == ra_2_Distributed))

      {

        // test case 7.1.7.1.3



//5MhzSupport

        if (v_N_RBDL == 25)

          {

            v_Max_nPRB := 24;

          }else {  v_Max_nPRB := 16; }
      }

    else

      {

        FatalError(__FILE__, __LINE__, "Step 1: invalid DCI format and Resource allocation combination ");

      }

    f_ConfigureULGrant_MaxTBSupported(v_UE_Category, f_ConvertUL_BandwidthToInteger(f_EUTRA_CellInfo_GetUL_ChBandwidth(eutra_Cell1)));

    for (v_N_PRB := 1; v_N_PRB <= v_Max_nPRB; v_N_PRB := v_N_PRB + 1) {  //for nPRB 1 to v_Max_nPRB

      // For DCI 1 and RA 0, not all 1..v_Max_nPRB are applicable, needs to be addressed

      if ((p_Format == dci_1) and (p_ResourceAllocType == ra_0)) {

        //5MhzSupport



 if(v_N_RBDL == 25)      // 5 Mhz BW

          {

            // all PRBs applicable

          } else if (v_N_RBDL == 50)      // 10 Mhz BW

          {

            if ((v_N_PRB mod 3)==1) { continue; } // not applicable

          }

        else if (v_N_RBDL== 75)  // 15 Mhz BW

          {

            if (((v_N_PRB mod 4)==1) or ((v_N_PRB mod 4)==2)) { continue; } // not applicable

          }

        else if (v_N_RBDL == 100) // 20 Mhz BW

          {

            if ((v_N_PRB mod 4)!=0) { continue; } // not applicable

          }

        else

          {

            FatalError(__FILE__, __LINE__, "Step 1: invalid MAX Bandwidth set");

          }

      }

      for (v_I_MCS := 0; v_I_MCS <= 28 ; v_I_MCS := v_I_MCS + 1) {   //for iMCS 0 to 28

        //SS looks up iTBS in table 7.1.7.1-1 in TS 36.213 based on the value of iMCS

        v_I_TBS := f_EUTRA_Get_iTBS_For_iMCS(v_I_MCS);

        //SS looks up TBsize in table 7.1.7.2.1-1 in TS 36.213 based on values of iTBS and nPRB

        v_TBSize := f_EUTRA_LookUp_TB_Size(v_N_PRB, v_I_TBS);

        if (v_TBSize < 104)             //this results in zero number of PDCP SDUs

          // @sic R5-096421 R5s100321 R5s100422 sic@

          {

            continue ;

          }  // skip this nprb and Imcs combination.

        if (v_TBSize > f_EUTRA_GetMaxDLSCH_TB_Bits_Per_TTI(v_UE_Category))

          {

            continue;

          } ; // skip this nprb and Imcs combination.

        if ( not (fl_CalculateCodingRate(v_I_MCS, v_N_PRB, v_TBSize))) // @sic R5s100155 sic@

          {

            continue;

          }  // skip this nprb and Imcs combination.

        // Steps 2 to 4 are performed if  TBsize is less or equal to UE capability

        // "Maximum number of DL-SCH transport block bits received within a TTI" as specified in Table 7.1.7.1.1.3.2-1

        // When TBsize is less than 48 bits, no of PDCP SDUs results in 0

        //@siclog "Step 2" siclog@

        // SS creates one or more PDCP SDUs, depending on TBsize, in accordance with Table 7.1.7.1.1.3.2-2.

        f_EUTRA_Set_PDCP_SDUs_Size(v_N, v_PDCPsize, v_TBSize);  //calculates v_N,v_PDCPsize

        //@siclog "Step 3" siclog@

        // SS sends MAC PDU (NxPDCP SDUs)

        //Configure SS for DCI Format x with RA type y and a Resource block assignment (RBA) correspondent

        //to nPRB as specified in 7.1.6.1 in TS 36.213 and Modulation and coding scheme iMCS

        f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_CcchDcchDtchDL_Config_REQ(eutra_Cell1,               // @sic R5s110320 additional Changes sic@

                                                                               cs_TimingInfo_Now,

                                                                               cs_DciDLInfo_Explicit(cs_DciDlInfoExplicit(v_I_MCS,

                                                                                                                          p_Format,

                                                                                                                          p_ResourceAllocType,

                                                                                                                          v_N_PRB))));

        //@siclog "Step 3" siclog@

        // SS sends MAC PDU (NxPDCP SDUs)

        v_PDCP_SDUList := f_Build_N_PDCP_SDUList(v_N, v_PDCPsize);

        f_Tx_N_PDCP_SDUs(v_PDCP_SDUList);

        //@siclog "Step 4" siclog@

        // SS receives NxPDCP SDUs

        f_Rx_N_PDCP_SDUs(v_PDCP_SDUList);

      }//End for iMCS 0 to 28

    }//End for nPRB 1 to 110

    f_EUTRA_TestBody_Set(false);

    //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Post Amble

    //////////////////////////////////////////////////////////////////////////////////

    //@sic R5-103847 Deactivate test loop and open test loop is removed sic@

    //Switch off UE and expect a DETACH REQUEST message

    f_EUTRA_Postamble(eutra_Cell1, E3_TESTMODE );//@sic R5-103875 sic@

  }//f_7_1_7_x

  /*

   * @desc      basic function for 7.1.7.x test cases

   * @param     p_Format

   * @param     p_ResourceAllocType

   * @param     p_CodeWordSwapFlag

   * @status    APPROVED (LTE)

   * @status

   */

  function f_EUTRA_7_1_7_1_x_MIMO (PdcchDciFormat_Type p_Format,

                                   PdcchResourceAllocation_Type p_ResourceAllocType,

                                   B1_Type p_CodeWordSwapFlag ) runs on EUTRA_PTC

  { /* L2/MAC/7.1

     * 7.1.7.x : DL-SCH Transport Block Size selection

     */

    //@sic R5-113849 sic@

    //@sic R5s110582 R5s110585 sic@

    var integer v_UE_Category ;

    var integer v_TBSize1:=0;   // transport block size in bits

    var integer v_TBSize:=0;    // transport block size in bits

    var integer v_I_MCS:=0;     // Modulation and Coding Scehme Index

    var integer v_I_TBS:=0;     // Transport Block Size Index

    var integer v_N_PRB:=1;     // Number of physical resource Blocks

    var integer v_N_RBDL := 100; // Species the DL Bandwidth.

    var integer v_Max_nPRB;     // to store the max allowed PRB resources that can be allocated

    var integer v_PDCPsize:=0;  // PDCP SDU Size

    var integer v_N:=1;         // Number of PDCP SDU

    var PDCP_SDUList_Type v_PDCP_SDUList := {''O};

    ///////////////////////////////////////////////////////////////////////////////////

    //      PREAMBLE

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_Init (c1);

    f_EUTRA_CellInfo_SetHighestBandwidth(eutra_Cell1);

    f_EUTRA_CellInfo_SetMIMO_2TX(eutra_Cell1); //@sic R5s110582 R5s110585 sic@

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeA);

    v_UE_Category := f_EUTRA_MobileInfo_GetUECapability_UECategoryExtn(); //@sic R5s120641 R5-126021 sic@

    f_CheckUECategory(v_UE_Category);

    // Bring UE to state Loopback Activated (state 4)

    f_EUTRA_LoopBackActivation_State4_717_MIMO(); //@sic R5s110697 sic@

    /* f_EUTRA_SS_CommonCellConfig(eutra_Cell1,

                                cas_CellModify_REQ_PDSCH_Pwr_Ratio_TBS_Test(eutra_Cell1,

                                                                            cs_TimingInfo_Now));*/

    ///////////////////////////////////////////////////////////////////////////////////

    //      Test Body

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_TestBody_Set(true);

    //Steps 1 to 4 are repeated for values of nPRB  from 1 to v_Max_nPRB and iMCS from 0 to 28

    v_N_RBDL := f_ConvertDL_BandwidthToInteger(f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1));

    if ((p_Format == dci_2A) and (p_ResourceAllocType == ra_0))

      {

        // test case 7.1.7.1.5

        v_Max_nPRB := v_N_RBDL;

        // MAX index can be NRBDL but not all nPRB's allowed

      }

    else if ((p_Format == dci_2A) and (p_ResourceAllocType == ra_1))

      {

        // test case 7.1.7.1.2

        //5MhzSupport

  if (v_N_RBDL == 25)

          {

            v_Max_nPRB := 11;

          }

        else  if (v_N_RBDL == 50)

          {

            v_Max_nPRB := 14;

          }

        else if (v_N_RBDL == 75)

          {

            v_Max_nPRB := 16;

          }

        else if (v_N_RBDL == 100)

          {

            v_Max_nPRB := 22;

          }

        else

          {

            FatalError(__FILE__, __LINE__, "Step 1: invalid MAX Bandwidth set");

          }

      }

    else

      {

        FatalError(__FILE__, __LINE__, "Step 1: invalid DCI format and Resource allocation combination ");

      }

    f_ConfigureULGrant_MaxTBSupported(v_UE_Category, f_ConvertUL_BandwidthToInteger(f_EUTRA_CellInfo_GetUL_ChBandwidth(eutra_Cell1)));

    for (v_N_PRB := 1; v_N_PRB <= v_Max_nPRB; v_N_PRB := v_N_PRB + 1) {  //for nPRB 1 to v_Max_nPRB

      // For DCI 2A and RA 0, not all 1..v_Max_nPRB are applicable, needs to be addressed

      if ((p_Format == dci_2A) and (p_ResourceAllocType == ra_0)) {

        //5MhzSupport

     if (v_N_RBDL == 25)      // 5 Mhz BW

          {

             // all PRBs applicable

          } else if (v_N_RBDL == 50)      // 10 Mhz BW

          {

            if ((v_N_PRB mod 3)==1) { continue; } // not applicable

          }

        else if (v_N_RBDL== 75)  // 15 Mhz BW

          {

            if (((v_N_PRB mod 4)==1) or ((v_N_PRB mod 4)==2)) { continue; } // not applicable

          }

        else if (v_N_RBDL == 100) // 20 Mhz BW

          {

            if ((v_N_PRB mod 4)!=0) { continue; } // not applicable

          }

        else

          {

            FatalError(__FILE__, __LINE__, "Step 1: invalid MAX Bandwidth set");

          }

      }

      for (v_I_MCS := 0; v_I_MCS <= 28 ; v_I_MCS := v_I_MCS + 1) {   //for iMCS 0 to 28

        //SS looks up iTBS in table 7.1.7.1-1 in TS 36.213 based on the value of iMCS

        v_I_TBS := f_EUTRA_Get_iTBS_For_iMCS(v_I_MCS);

        //SS looks up TBsize in table 7.1.7.2.1-1 in TS 36.213 based on values of iTBS and nPRB

        v_TBSize1 := f_EUTRA_LookUp_TB_Size(v_N_PRB, v_I_TBS);

        if ( not (fl_CalculateCodingRate(v_I_MCS, v_N_PRB, v_TBSize1,2))) // Coding Rate exceeds

          {

            continue;

          }  // skip this nprb and Imcs1 combination.

        if (v_TBSize1 > f_EUTRA_GetMaxDLSCH_TB_Bits_Per_TTI(v_UE_Category))

          {

            continue;

          } ; // skip this nprb and Imcs1  combination.

        v_TBSize := v_TBSize1 + v_TBSize1 ;

        if (v_TBSize < 136)             //this results in zero number of PDCP SDUs

          {

            continue ;

          }  // skip this nprb and Imcs1 and Imcs2 combination.

        if (v_TBSize > f_EUTRA_GetMaxDLSCH_2TB_Bits_Per_TTI_MIMO(v_UE_Category))

          {

            continue;

          } ; // skip this nprb and Imcs1 & Imcs2 combination.

        // Steps 2 to 4 are performed if the sum of the sizes of TBsize#1  and TBsize#2 is less than or equal to

        //UE capability "Maximum number of DL-SCH transport block bits received within a TTI" as specified in Table 7.1.7.1.5.3.2-1

        //and larger than or equal to 72 136 bits as specified in Table 7.1.7.1.5.3.2-2, and the effective channel code rate, as defined

        //in TS 36.213 clause 7.1.7, is lower than or equal to 0.930.

        //@siclog "Step 2" siclog@

        // SS creates one or more PDCP SDUs, depending on TBsize, in accordance with Table 7.1.7.1.5/6.3.2-2.

        f_EUTRA_Set_PDCP_SDUs_Size_MIMO(v_N, v_PDCPsize, v_TBSize);  //calculates v_N,v_PDCPsize

        //@siclog "Step 3" siclog@

        // SS sends MAC PDU (NxPDCP SDUs)

        //Configure SS for DCI Format x with RA type y and a Resource block assignment (RBA) correspondent

        //to nPRB as specified in 7.1.6.1 in TS 36.213 and Modulation and coding scheme iMCS

        f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_CcchDcchDtchDL_Config_REQ(eutra_Cell1,

                                                                               cs_TimingInfo_Now,

                                                                               cs_DciDLInfo_Explicit(cs_DciDlInfoExplicit_MIMO(cs_Imcs_Value(v_I_MCS),

                                                                                                                               cs_Imcs_Value(v_I_MCS),

                                                                                                                               p_Format,

                                                                                                                               p_ResourceAllocType,

                                                                                                                               v_N_PRB,

                                                                                                                               -,

                                                                                                                               -,

                                                                                                                               p_CodeWordSwapFlag))));

        //@siclog "Step 3" siclog@

        // SS sends MAC PDU (NxPDCP SDUs)

        v_PDCP_SDUList := f_Build_N_PDCP_SDUList_MIMO(v_N, v_PDCPsize);

        f_Tx_N_PDCP_SDUs(v_PDCP_SDUList);

        //@siclog "Step 4" siclog@

        // SS receives NxPDCP SDUs

        f_Rx_N_PDCP_SDUs(v_PDCP_SDUList);

      }//End for iMCS 0 to 28

    }//End for nPRB 1 to 110

    f_EUTRA_TestBody_Set(false);

    //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Post Amble

    //////////////////////////////////////////////////////////////////////////////////

    //@sic R5-103847 Deactivate test loop and open test loop is removed sic@

    // configure SS in normal grant transmission

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1,

                                cas_CcchDcchDtchDL_Config_REQ(eutra_Cell1,

                                                              cs_TimingInfo_Now,

                                                              cs_DciDLInfo_Auto(cs_DciInfo_CcchDcchDtchDL(f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1))))); //@sic R5s110320 sic@

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1);

    //Switch off UE and expect a DETACH REQUEST message

    f_EUTRA_Postamble(eutra_Cell1, E3_TESTMODE );//@sic R5-103875 sic@

  }//f_7_1_7_x_MIMO_EUTRA


